# UNIVERSIDAD TECNOLÓGICA NACIONAL
## FACULTAD REGIONAL ROSARIO

CÁTEDRA DE “Balances de Materia y Energía”




# Balances de Materia y Energía en estado NO Estacionario

MMXXV


Prof. Adj.: Ph.D. Néstor Hugo Rodríguez
Aux 1°: Ing. Mabel Andrea Dupuy

# 1- Balances de materia y energía en estado estacionario

## a-Un componente




BM) $$Fs = F1 + F2$$

BE) $$Fs \times H_S = F1 \times H_1 + F2 \times H_2$$

$$H_1 = a \times T_1 + b$$
$$H_2 = a \times T_2 + b$$
$$H_S = a \times T_S + b$$

Ejemplo de Aplicación

Sean dos corrientes de un componente puro con las siguientes características, calcular el flujo y temperatura de salida.

F1= 1 [kg/s]
T1= 363.15 [ºK]

F2= 2 [kg/s]
T2= 293.15 [ºK]

a= 42.071 [KCal/ Kg ºK]
b= -17069 [KCal/ Kg]

***Resolución algebraica:***

$$Fs = F1 + F2 = 1\ [Kg/s]\ + 2\ [Kg/s] = 3\ [Kg/s]$$
$$H_1 = a \times 363{,}15 + b = 1{,}526 \times 10^7\ [Kcal/kg]$$
$$H_2 = a \times 293{,}15 + b = 1{,}232 \times 10^7\ [Kcal/kg]$$

$$H_S = \frac{F1 \times H_1 + F2 \times H_2}{Fs} = \frac{1[Kg/s] \times 1{,}526 \times 10^7 [Cal/Kg] + 2[Kg/s] \times 1{,}232 \times 10^7 [Cal/Kg]}{3[Kg/s]} = 1{,}330 \times 10^7 [Cal/Kg]$$

$$T_S = \frac{H_S - b}{a} = \frac{1{,}330 \times 10^7 [Kcal/Kg] - (-17069)[Kcal/Kg]}{42{,}071 [Kcal/Kg\,^{\circ}K]} = 316{,}483 [^{\circ}\mathrm{K}]$$

### *Resolución con simulador EMSO*

```
#* Balance *#

using "types";

FlowSheet Balance

PARAMETERS
F1 as Real (Brief="flujo masico entrada 1",Unit='kg/s');
F2 as Real (Brief="flujo masico entrada 2",Unit='kg/s');
T1 as Real(Brief="temp entrada 1", Unit='K');
T2 as Real(Brief="temp entrada 2", Unit='K');
a as Real(Brief="param entalpia masa",Unit='kg/kg/K');
b as Real(Brief="param entalpia masa",Unit='kg/kg');


VARIABLES
Fs as Real(Brief="flujo masico salida",Unit='kg/s');
Ts as Real(Brief="temp salida", Unit='K');

EQUATIONS
Fs=F1+F2;
(a*Ts+b)*Fs=(a*T1+b)*F1+(a*T2+b)*F2;


SET
F1=2*'kg/s';
T1=363.15*'K';
F2=1*'kg/s';
T2=293.15*'K';
a=4.2071*'kg/kg/K’;
b=- 17069*'kg/kg’;

INITIAL
Fs=1*'kg/h';

GUESS
Fs=0*'kg/s';

OPTIONS
TimeStart=0;
TimeEnd=100;
TimeStep=1;
TimeUnit='h';
DAESolver(File="dassl");
Dynamic=false;

End
```

Resultados:
Fs=  3 [Kg/s]
Ts= 316,483 [ºK]

### *Resolución con simulador Hysys™*



### *Resolución con planillas Excel*

Implementar el siguiente modelo:

| | A | B | C | D | E |
|---|---|---|---|---|---|
| 1 | Corriente | F1 | F2 | _ | Fs |
| 2 | Flujo masico | 1 | 2 | | =B2+C2 |
| 3 | Densidad | 11,18568232 | 9,36329588 | | =E4*B3+E5*C3 |
| 4 | Temperatura | 363,15 | 293,15 | | =(E7-E6)/E5 |
| 5 | a | 42071 | 42071 | | 42071 |
| 6 | b | -17069 | -17069 | | -17069 |
| 7 | Entalpia | =B5*B4+B6 | =C5*C4+C6 | | =(B2*B7+C2*C7)/E2 |

La que resuelta queda:

| | A | B | C | D | E |
|---|---|---|---|---|---|
| 1 | Corriente | F1 | F2 | _ | Fs |
| 2 | Flujo masico | 1 | 2 | | 3 |
| 3 | Densidad | 11,186 | 9,363 | | 397463,303 |
| 4 | Temperatura | 363,150 | 293,150 | | 316,483 |
| 5 | a | 42.071 | 42.071 | | 42.071 |
| 6 | b | -17069 | -17069 | | -17069 |
| 7 | Entalpia | 1,526E+07 | 1,232E+07 | | 1,330E+07 |

## *Resolución método gráfico:*

Vemos que la entalpía sigue una tendencia lineal. Si dibujamos dicha línea y buscamos los puntos correspondientes a las temperaturas de 293,15 [ºK] y 363,15 [ºK] habremos hallado las entalpias de las corrientes F2 y F1 respectivamente. Dividiendo dicha línea en dos segmentos con una relación de 2 [Kg/s]/ 1 [Kg/s] pero en sentido opuesto (regla de la inversa de la palanca) habremos hallado Fs que corresponde precisamente a una temperatura de 316 [ºK]:



## b-Dos componentes:

Sea un flowsheet siguiente formado por dos corrientes puras de componentes 1 y 2, Sean F1, F2, T1 y T2 los flujos molares y temperaturas de ambas alimentaciones. Calcular la temperatura de salida:




BM) $$Fs = F1 + F2$$

BMxC) $$Fs \times x_B = F1$$
$$Fs \times x_T = F2$$

BE) $$Fs \times H_S = F1 \times H_1 + F2 \times H_2$$

$$H_1 = a_1 \times T_1 + b_1$$
$$H_2 = a_2 \times T_2 + b_2$$
$$H_S = \frac{F1 \times (a_1 \times T_S + b_1) + F2 \times (a_2 \times T_S + b_2)}{F_S}$$

Ejemplo de aplicación:

Benceno

F1= 30 [Kgmol/h]

T1= 293,15 [ºK]

a1= 133.2962 [KJ/Kg mol ºK]

b1= 9 691,2069 [KJ/Kg mol]

Tolueno

F2= 70 [Kgmol/h]

T2=343.15 [ºK]

a2= 160.5239 [KJ/Kg mol ºK]

b2= - 35 832.1603 [KJ/Kg mol]

### *Resolución algebraica:*

$$Fs = F1 + F2 = 30\ [Kgmol\ /\ h]\ + 70\ [Kgmol\ /\ h\ ] = 100\ \ [Kgmol\ /\ h\ ]$$
$$H_1 = a \times 363{,}15 + b = 4.8767 \times 10^4\ \ [KJ\ /\ kgmol]$$
$$H_2 = a \times 343{,}15 + b = 1.9252 \times 10^4 [KJ\ /\ kgmol]$$

$$H_S = \frac{F1 \times (a_1 \times T_S + b_1) + F2 \times (a_2 \times T_S + b_2)}{F_S}$$
$$H_S \times F_S = F1 \times a_1 \times T_S + F1 \times b_1 + F2 \times a_2 \times T_S + F2 \times b_2 = T_S \times (F1 \times a_1 + F2 \times a_2) + F1 \times b_1 + F2 \times b_2$$
$$T_S = \frac{H_S \times F_S - F1 \times b_1 - F2 \times b_2}{(F1 \times a_1 + F2 \times a_2)}$$

$$= \frac{30 [Kgmol / h] \times 4{,}8767 \times 10^4 [KJ\ /\ Kgmol] + 70 [Kgmol / h] \times 1{,}9252 \times 10^7 [KJ\ /\ Kgmol]}{100 [Kgmol / h]} = 2{,}8106 \times 10^4 [KJ\ /\ Kgmol]$$

$$T_S = \frac{2{,}8106 [KJ\ /\ Kgmol] \times 100 [Kgmol\ /\ h] - 30 [Kgmol\ /\ h] \times 9691 [KJ\ /\ Kgmol] + 70 [Kgmol\ /\ h] \times 35832 [KJ\ /\ Kgmol]}{30 [Kgmol\ /\ h] \times 133{,}29 [KJ\ /\ Kgmol\ ^\circ K] + 70 [Kgmol\ /\ h] \times 160{,}52 [KJ\ /\ Kgmol\ ^\circ K]}$$

$$T_S = 330{,}02 [^\circ K]$$



60000
50000
40000
30000
20000
10000
0
Entalpía molar [KJ/Kg]
273
278
283
288
293
298
303
308
313
318
323
328
333
338
343
Temperatura [ºK]
Benceno
Tolueno
Mezcla
7
3

*Resolución con simulador EMSO*

```
#* Balance *#

using "types";

FlowSheet Balance

PARAMETERS
F1 as Real(Brief="flujo masico entrada 1", Unit='kmol/h');
F2 as Real(Brief="flujo masico entrada 2", Unit='kmol/h');
T1 as Real(Brief="temp entrada 1", Unit='K');
T2 as Real(Brief="temp entrada 2", Unit='K');
a(2) as Real(Brief="param entalpia molar",Unit='kJ/kmol/K');
b(2) as Real(Brief="param entalpia molar",Unit='kJ/kmol/K');

VARIABLES
Fs as Real(Brief="flujo masico salida", Unit='kmol/h');
Ts as Real(Brief="temp salida", Unit='K');
x(2) as Real(Brief="comp salida");
H1 as Real(Brief="Entalpia molar 1",Unit='kJ/kmol');
H2 as Real(Brief="Entalpia molar 1",Unit='kJ/kmol');
Hs as Real(Brief="Entalpia molar 1",Unit='kJ/kmol');

EQUATIONS
Fs=F1+F2;
Fs*x(1)=F1;
Fs*x(2)=F2;
H1=a(1)*T1+b(1);
H2=a(2)*T2+b(2);
Hs*Fs=H1*F1+H2*F2;
Hs=x(1)*(a(1)*Ts+b(1))+x(2)*(a(2)*Ts+b(2));


SET
F1=30*'kmol/h';
T1=293.15*'K';
a(1)=133.2962*'kJ/kmol/K';
b(1)=9691.2069*'kJ/kmol';

F2=70*'kmol/h';
T2=343.15*'K';
a(2)=160.5239*'kJ/kmol/K';
b(2)=-35832.1603*'kJ/kmol';
```

```
INITIAL
Fs=0*'kmol/h';

GUESS
Fs=0*'kmol/h';

OPTIONS
TimeStart=0;
TimeEnd=100;
TimeStep=1;
TimeUnit='h';
DAESolver(File="dassl");
Dynamic=false;

end
```

Resultados:

Fs= 100 [Kgmol/h]

Ts= 330,026 [ºK]

$x_S$=[0,3 ; 0,7]

## *Resolución con simulador Hysys™*



### *Resolución con planillas Excel*

Implementar el siguiente modelo:

| | A | B | C | D | E |
|---|---|---|---|---|---|
| 1 | Corriente | F1 | F2 | | Fs |
| 2 | Flujo | 30 | 70 | | =B2+C2 |
| 3 | Temperatura | 293.15 | 343.15 | | =(E6*E2-B2*B5-C2*C5)/(B2*B4+C2*C4) |
| 4 | a | 133.2962 | = 160.5239 | | |
| 5 | b | 9691.2069 | - 35832.1603 | | |
| 6 | H | =B4*B3+B5 | =C4*C3+C5 | | =(B6*B2+C6*C2)/E2 |
| 7 | x1 | 1 | 0 | | =B2/E2 |
| 8 | x2 | 0 | 1 | | =C2/E2 |

Que queda así:

| | A | B | C | D | E |
|---|---|---|---|---|---|
| 1 | Corriente | F1 | F2 | | Fs |
| 2 | Flujo | 30 | 70 | | 100 |
| 3 | Temperatura | 293.15 | 343.15 | | 330.03 |
| 4 | a | 133.30 | 160.52 | | |
| 5 | b | 9.6912E+03 | -3.5832E+04 | | |
| 6 | H | 4.8767E+04 | 1.9252E+04 | | 2.8106E+04 |
| 7 | x1 | 1.000 | 0.000 | | 0.300 |
| 8 | x2 | 0.000 | 1.000 | | 0.700 |

*Resolución método gráfico:*

Vemos que la entalpía sigue una tendencia lineal. Si dibujamos dicha línea y buscamos los puntos correspondientes a las temperaturas de 293,15 [ºK] y 343,15 [ºK] habremos hallado las entalpias de las corrientes F2 y F1 respectivamente. Dividiendo dicha línea en dos segmentos con una relación de 7:3 pero en sentido opuesto (regla de la inversa de la palanca) habremos hallado Fs que corresponde precisamente a una temperatura de 328 [ºK]:




La diferencia entre el método gráfico y el algebraico es de 1,87 [ºK]

## 2- Balances de materia y energía en estado dinámico

### a-Un componente

Sea el flowsheet formado por un tanque cilíndrico de altura HT, y diámetro DT con un orificio en la parte infe líquido (S). Dos corriente puras diferentes ingresan por la parte superior con flujos molares F1 y F2 y temperatu la altura (h), composición y flujo de salida (Fs) y su temperatura en función del tiempo cuando al principio est componente 1.




$$\boxed{\frac{dh}{dt} = \frac{(F1 + F2 - Fs)}{\rho_S \times AT}}$$

$$\boxed{\begin{array}{l} Ao = \dfrac{\pi \times Do^2}{4} \\ AT = \dfrac{\pi \times DT^2}{4} \end{array}}$$

$$\boxed{Fs = \rho_S \times Ao \times \sqrt{2 \times g \times h}}$$

$$\frac{d(H_S \times M)}{dt} = F1 \times H_1 + F2 \times H_2 - Fs \times H_S$$

$$\frac{d(H_S \times h \times AT \times \rho_S)}{dt} = F1 \times H_1 + F2 \times H_2 - Fs \times H_S$$

$$AT \times \rho_S \times \frac{d(H_S \times h)}{dt} = F1 \times H_1 + F2 \times H_2 - Fs \times H_S$$

$$\frac{d(H_S \times h)}{dt} = \frac{F1 \times H_1 + F2 \times H_2 - Fs \times H_S}{AT \times \rho_S}$$

$$h\frac{d(H_S)}{dt} + H_S \frac{d(h)}{dt} = \frac{F1 \times H_1 + F2 \times H_2 - Fs \times H_S}{AT \times \rho_S}$$

$$\boxed{\frac{d(H_S)}{dt} = \left[\frac{F1 \times H_1 + F2 \times H_2 - Fs \times H_S}{AT \times \rho_S} - H_S \frac{d(h)}{dt}\right] \Big/ h}$$

$$\boxed{\begin{gathered} H_1 = a_1 \times T_1 + b_1 \\ H_2 = a_2 \times T_2 + b_2 \\ H_S = \frac{F1 \times (a_1 \times T_S + b_1) + F2 \times (a_2 \times T_S + b_2)}{F_S} \end{gathered}}$$

Cuando dh/dt  se hace 0, se alcanza la altura de estado estacionario, esta puede ser máxima o minina según si el arranque f
A su vez, cuando dHs/dt se hace cero alcanza la temperatura de equilibrio

$$\frac{dh}{dt} = \frac{\left(F1 + F2 - \rho_S \times Ao \times \sqrt{2 \times g \times h_{EE}}\right)}{\rho_S \times AT} = 0$$

$$\rho_S \times Ao \times \sqrt{2 \times g \times h_{EE}} = F1 + F2$$

$$\sqrt{2 \times g \times h_{EE}} = \frac{F1 + F2}{\rho_S \times Ao}$$

$$2 \times g \times h_{EE} = \left(\frac{F1 + F2}{\rho_S \times Ao}\right)^2$$

$$\boxed{h_{EE} = \frac{\left(\frac{F1 + F2}{\rho_S \times Ao}\right)^2}{2 \times g}}$$

Ejemplo de aplicación:




Parámetros del problema:

| | | |
|---|---|---|
| F1: | 0,50 | [kmol/s] |
| T1= | 363,15 [ºK] | |
| ρ1: | 54 | [mol/m$^3$] |
| a1= | 42,071 [KJ/Kg mol ºK] | |
| b1= | -17069 [KJ/Kg mol] | |
| | | |
| F2: | 0,30 | [mol/s] |
| ρ1: | 54 | [mol/m$^3$] |
| T2= | 293,15 [ºK] | |
| a2= | 42,071 [KJ/Kg mol ºK] | |
| b2= | -17069 [KJ/Kg mol] | |
| | | |
| DT: | 4 | [m] |
| Do: | 0.05 | [m] |
| HT: | 5 | [m] |
| AT: | 12,566 [m$^2$] | |
| Ao: | 1,9635E-03 [m$^2$] | |

Condiciones iniciales:

Ts = 300 ºK → HS= -4447,7 KJ/Kgmol

h = 2 [m]

***Resolución algebraica a tiempo infinito (estado estacionario):***

$$Fs = F1 + F2 = 0{,}5\ [Kgmol/s] + 0{,}3\ [Kgmol/s] = 0{,}8\ [Kgmol/s]$$

$$H_1 = a \times 363{,}15 + b = \text{-}1{,}791 \times 10^3 \ [KJ / kgmol]$$
$$H_2 = a \times 293{,}15 + b = \text{-}4{,}763 \times 10^3 [KJ / kgmol]$$

$$H_S = \frac{F1 \times (a \times T_S + b) + F2 \times (a \times T_S + b)}{F_S}$$
$$H_S \times F_S = F1 \times a \times T_S + F1 \times b + F2 \times a \times T_S + F2 \times b = T_S \times (F1 \times a + F2 \times a) + F1 \times b + F2 \times b$$
$$H_S = a \times T_S + b$$
$$T_S = \frac{H_S - b}{a}$$
$$T_S = \frac{(-2{,}895 \times 10^3)\left[\frac{KJ}{Kgmol}\right] - (-17069)\left[\frac{KJ}{Kgmol}\right]}{42{,}071\left[\frac{KJ}{Kgmol \times {}^\circ K}\right]}$$

$$T_S = 336{,}90[{}^\circ K]$$

### *Método de Euler.*

Sin entrar en detalles podemos deducir éste algoritmo a partir de la definición de derivada:

$$f'(x) = \lim_{\Delta x \to 0} \frac{f(x + \Delta x) - f(x)}{\Delta x}$$

Tomando un Δx suficientemente chico eliminamos el lim y de ese modo obtenemos una aproximación a la derivada:

$$f'(x) = \frac{f(x + \Delta x) - f(x)}{\Delta x}$$

$$f(x + \Delta x) = \Delta x \times f'(x) + f(x)$$

De ese modo, sabiendo la derivada y partiendo de condiciones iniciales conocidas (ecuaciones derivadas ordinarias o EDO) se puede calcular el valor próximo de la variable. Este valor es el que se usará para el tiempo siguiente e iterativamente se obtiene toda la evolución de la función que se busca. Cabe aclarar que la solución no será una curva sino una serie continua de trazos.

Formalmente se obtiene de la expansión de Taylor truncando las derivadas superiores a 1. Los términos eliminados (residuo) constituyen el error que se comete.

Desde el punto de vista práctico cuanto más pequeño es el salto Δx más se aproxima la solución numérica a la solución algebraica como se aprecia en la figura. Por el contrario, a pasos grandes se comete más error el cual se propaga en el tiempo haciendo que ambas soluciones diverjan rápidamente.

El ejemplo está hecho sobre la función $f'(x) = 3 \times x^2$ cuya solución algebraica es $f(x) = x^3 + C_0$




En la figura se aprecian la curva derivada de la solución algebraica y 3 soluciones numéricas para 3 diferentes Δx (1, 5 y 10). Aún cuando en la figura no llegue a apreciarse las soluciones numéricas no son curvas sino tramos rectos pero dadas las suavidades de las misma casi no se percibe.

Así pues en estos casos siempre nos enfrentaremos al dilema ya que si bien Δx pequeños ajustan mejor el número de pasos a realizar se incrementa. Los algoritmos más eficientes hacen usos de pasos variables, pequeño en los casos de grandes variaciones de la curva y grande cuando la misma empieza a "plancharse".

### *Sistemas de ecuaciones:*

El algoritmo es similar pero ahora se aplica todas las ecuaciones diferenciales simultáneamente. A modo de ejemplo se aplicará al movimiento acelerado. Las ecuaciones que representan al fenómeno son:

$$\left(\frac{dx}{dt}\right) = v$$
$$\left(\frac{dv}{dt}\right) = a$$

Para el caso en que la aceleración es constante (1 $m/s^2$) y los valores iniciales (a t=0), tanto la velocidad como el desplazamiento son nulos.

Recordemos sus soluciones algebraicas:

$v = a \times t + v_0$

$x = \frac{1}{2} a \times t^2 + x_0$









### *Resolución con simulador EMSO movimiento acelerado*

```
#* Ejemplo sistemas de ecuaciones*#
using "types";

FlowSheet Movimiento

PARAMETERS
	a as Real;

VARIABLES
	x as Real;
	v as Real;

EQUATIONS
	diff(x)=v;
	diff(v)=a;

SET
	a=1;

INITIAL
	x=0;
	v=0;

GUESS

OPTIONS
TimeStart=0;
TimeEnd=10;
TimeStep=0.1;
TimeUnit='s';
```

```
DAESolver(File="dassl");
Dynamic=true;

end
```

*Resolución con simulador EMSO tanque.*

```
#* Balance *#

using "types";

FlowSheet Balance

PARAMETERS
F1 as Real (Brief="flujo molar entrada 1",Unit='kmol/s');
F2 as Real (Brief="flujo molar entrada 2",Unit='kmol/s');
DT as Real (Brief="Diametro tanque", Unit='m');
g as Real (Brief="Ac de la gravedad",Unit='m/s^2', Default=9.8);
Do as Real (Brief="Diametro orificio",Unit='m');
pi as Real (Brief="Cte pi", Default=3.1415923536);
Ro as Real(Brief="densidad molar",Unit='kmol/m^3');
T1 as Real (Brief="Temperatura entrada 1",Unit='K');
T2 as Real (Brief="Temperatura entrada 2",Unit='K');
a as Real(Brief="param entalpia molar",Unit='kJ/kmol/K');
b as Real(Brief="param entalpia molar",Unit='kJ/kmol');

VARIABLES
Fs as Real (Brief="flujo molar salida",Unit='kmol/s');
h as Real(Brief="altura",Unit='m');
H1 as Real (Brief="entalpia entrada 1",Unit='kJ/kmol');
H2 as Real (Brief="entalpia entrada 2",Unit='kJ/kmol');
Hs as Real (Brief="entalpia salida",Unit='kJ/kmol');
Ts as Real (Brief="Temperatura salida",Unit='K');
AT as Real (Brief="Area tanque", Unit='m^2');
Ao as Real (Brief="Area orificio", Unit='m^2');

EQUATIONS
AT=pi*DT^2/4;
Ao=pi*Do^2/4;
diff(h)=(F1+F2-Fs)/(AT*Ro);
Fs=Ro*Ao*sqrt(2*g*h);
diff(h*Hs)=(F1*H1+H2*F2-Hs*Fs)/(Ro*AT);
H1=a*T1+b;
H2=a*T2+b;
Hs=a*Ts+b;
```

```
SET
F1=0.5*'kmol/s';
F2=0.3*'kmol/s';
T1=363.15*'K';
T2=293.15*'K';
a=42.071*'kJ/kmol/K';
b=-17069*'kJ/kmol';
Ro=54*'kmol/m^3';
DT=4*'m';
Do=0.05*'m';

INITIAL
h=1*'m';
#Ts=310*'K';
Hs=-4447.7*'kJ/kmol';

GUESS

OPTIONS
TimeStart=0;
TimeEnd=20000;
TimeStep=10;
TimeUnit='s';
DAESolver(File="dassl");
Dynamic=true;

end
```









En el estado estacionario (dynamic=false):

Fs= 0,8 [kgmol/s]

h= 2,90449 [m]

Ts= 336,9 [ºK]

### *Resolución con planillas Excel*

Implementar el siguiente modelo para su resolución en estado estacionario:

| | A | B | C | D | E | F |
|---|---|---|---|---|---|---|
| 1 | Corriente | F1 | F2 | | Fs | Unidades |
| 2 | Flujo masico | 0,5 | 0,3 | | =B2+C2 | Kgmol/s |
| 3 | Densidad | 54 | 54 | | =54 | Kgmol/m3 |
| 4 | Temperatura | 363,15 | 293,15 | | =(E7-E6)/E5 | ºK |
| 5 | A | 42,071 | 42,071 | | 42,071 | KJ/Kgmol ºK |
| 6 | B | -17069 | -17069 | | -17069 | KJ/Kgmol |
| 7 | Entalpia | =$B$5*B4+$B$6 | =C5*C4+C6 | | =(B2*B7+C2*C7)/E2 | KJ/Kgmol |

| | H | I | J |
|---|---|---|---|
| 1 | Param | Valor | Unidades |
| 2 | Ro | 54 | kmol/m^3 |
| 3 | Do | 0,05 | m |
| 4 | Ao | =PI()*I3*I3/4 | $m^2$ |
| 5 | DT | 4 | m |
| 6 | AT | =PI()*I5*I5/4 | $m^2$ |
| 7 | Hee | =((E2/(I2*I4))^2)/(2*9,8) | m |

Que resuelto en EE queda:

| | A | B | C | D | E | F | G | H | I | J |
|---|---|---|---|---|---|---|---|---|---|---|
| 1 | Corriente | F1 | F2 | | Fs | Unidades | | Param | Valor | Unidades |
| 2 | Flujo masico | 0,50000 | 0,30000 | | 0,8 | Kgmol/s | | Ro | 54 | kmol/m^3 |
| 3 | Densidad | 54,000 | 54,000 | | 54,000 | Kgmol/m3 | | Do | 0,05 | m |
| 4 | Temperatura | 363,150 | 293,150 | | 336,900 | ºK | | Ao | 1,9635E-03 | $m^2$ |
| 5 | A | 42,071 | 42,071 | | 42,071 | KJ/Kgmol ºK | | DT | 4 | m |
| 6 | B | -17069 | -17069 | | -17069 | KJ/Kgmol | | AT | 12,566 | $m^2$ |
| 7 | Entalpia | -1,791E+03 | -4,736E+03 | | -2,895E+03 | KJ/Kgmol | | Hee | 2,9045 | m |

Para obtener la solución dinámica adicionar lo siguiente a la planilla anterior:

| | A | B | C | D | E | F | G | H |
|---|---|---|---|---|---|---|---|---|
| 8 | | | | | | | | |
| 9 | Dt | 70 | | | | hf | =B300 | Fsf |
| 10 | | | | | | | | |
| 11 | T | h | F1 | T1 | H1 | F2 | T2 | H2 |
| 12 | 0 | 1 | =B2 | =$B$4 | =D12*$B$5+$B$6 | =C2 | =$C$4 | =G12*$C$5+$C$6 |
| 13 | =A12+$B$9 | =B12+$B$9*L12 | =C12 | =D12 | =D13*$B$5+$B$6 | =F12 | =G12 | =G13*$C$5+$C$6 |
| 14 | =A13+$B$9 | =B13+$B$9*L13 | =C13 | =D13 | =D14*$B$5+$B$6 | =F13 | =G13 | =G14*$C$5+$C$6 |
| 15 | =A14+$B$9 | =B14+$B$9*L14 | =C14 | =D14 | =D15*$B$5+$B$6 | =F14 | =G14 | =G15*$C$5+$C$6 |
| 16 | =A15+$B$9 | =B15+$B$9*L15 | =C15 | =D15 | =D16*$B$5+$B$6 | =F15 | =G15 | =G16*$C$5+$C$6 |
| 17 | =A16+$B$9 | =B16+$B$9*L16 | =C16 | =D16 | =D17*$B$5+$B$6 | =F16 | =G16 | =G17*$C$5+$C$6 |
| 18 | =A17+$B$9 | =B17+$B$9*L17 | =C17 | =D17 | =D18*$B$5+$B$6 | =F17 | =G17 | =G18*$C$5+$C$6 |
| 19 | =A18+$B$9 | =B18+$B$9*L18 | =C18 | =D18 | =D19*$B$5+$B$6 | =F18 | =G18 | =G19*$C$5+$C$6 |
| 20 | =A19+$B$9 | =B19+$B$9*L19 | =C19 | =D19 | =D20*$B$5+$B$6 | =F19 | =G19 | =G20*$C$5+$C$6 |
| .. | ..... | ..... | .... | ..... | ....... | ...... | ...... | ................. |
| 300 | =A299+$B$9 | =B299+$B$9*L299 | =C299 | =D299 | =D300*$B$5+$B$6 | =F299 | =G299 | =G300*$C$5+$C$6 |

| | I | J | k |
|---|---|---|---|
| 8 | | | |
| 9 | =I300 | TSf | =K300 |
| 10 | | | |
| 11 | Fs | Hs | Ts |
| 12 | =$E$3*$I$4*RAIZ(2*9,8*B12) | -4447,7 | =(J12-$E$6)/$E$5 |
| 13 | =$E$3*$I$4*RAIZ(2*9,8*B13) | =J12+$B$9*M12 | =(J13-$E$6)/$E$5 |
| 14 | =$E$3*$I$4*RAIZ(2*9,8*B14) | =J13+$B$9*M13 | =(J14-$E$6)/$E$5 |
| 15 | =$E$3*$I$4*RAIZ(2*9,8*B15) | =J14+$B$9*M14 | =(J15-$E$6)/$E$5 |
| 16 | =$E$3*$I$4*RAIZ(2*9,8*B16) | =J15+$B$9*M15 | =(J16-$E$6)/$E$5 |
| 17 | =$E$3*$I$4*RAIZ(2*9,8*B17) | =J16+$B$9*M16 | =(J17-$E$6)/$E$5 |
| 18 | =$E$3*$I$4*RAIZ(2*9,8*B18) | =J17+$B$9*M17 | =(J18-$E$6)/$E$5 |
| 19 | =$E$3*$I$4*RAIZ(2*9,8*B19) | =J18+$B$9*M18 | =(J19-$E$6)/$E$5 |
| 20 | =$E$3*$I$4*RAIZ(2*9,8*B20) | =J19+$B$9*M19 | =(J20-$E$6)/$E$5 |
| ............ | ................ | ............. | .................. |
| 300 | =$E$3*$I$4*RAIZ(2*9,8*B300) | =J299+$B$9*M299 | =(J300-$E$6)/$E$5 |

| | L | M |
|---|---|---|
| 8 | | |
| 9 | | |
| 10 | | |
| 11 | dh | dHs |
| 12 | =(C12+F12-I12)/($E$3*$I$6) | =((C12*E12+F12*H12-I12*J12)/($I$6*$E$3)-J12*L12)/B12 |
| 13 | =(C13+F13-I13)/($E$3*$I$6) | =((C13*E13+F13*H13-I13*J13)/($I$6*$E$3)-J13*L13)/B13 |
| 14 | =(C14+F14-I14)/($E$3*$I$6) | =((C14*E14+F14*H14-I14*J14)/($I$6*$E$3)-J14*L14)/B14 |
| 15 | =(C15+F15-I15)/($E$3*$I$6) | =((C15*E15+F15*H15-I15*J15)/($I$6*$E$3)-J15*L15)/B15 |
| 16 | =(C16+F16-I16)/($E$3*$I$6) | =((C16*E16+F16*H16-I16*J16)/($I$6*$E$3)-J16*L16)/B16 |
| 17 | =(C17+F17-I17)/($E$3*$I$6) | =((C17*E17+F17*H17-I17*J17)/($I$6*$E$3)-J17*L17)/B17 |
| 18 | =(C18+F18-I18)/($E$3*$I$6) | =((C18*E18+F18*H18-I18*J18)/($I$6*$E$3)-J18*L18)/B18 |
| 19 | =(C19+F19-I19)/($E$3*$I$6) | =((C19*E19+F19*H19-I19*J19)/($I$6*$E$3)-J19*L19)/B19 |
| 20 | =(C20+F20-I20)/($E$3*$I$6) | =((C20*E20+F20*H20-I20*J20)/($I$6*$E$3)-J20*L20)/B20 |
| ….. | ……………………. | ……………………………. |
| 300 | =(C300+F300-I300)/($E$3*$I$6) | =((C300*E300+F300*H300-I300*J300)/($I$6*$E$3)-j300*L300)/B300 |

La que resuelta queda:

| | A | B | C | D | E | F | G | H | I | J | K | L | M |
|---|---|---|---|---|---|---|---|---|---|---|---|---|---|
| 8 | | | | | | | | | | | | | |
| 9 | Dt | 70 | | | | hf | 2,879 | Fsf | 0,796 | TSf | 336,90 | | |
| 10 | | | | | | | | | | | | | |
| 11 | T | h | F1 | T1 | H1 | F2 | T2 | H2 | Fs | Hs | Ts | dh | dHs |
| 12 | 0 | 1,000 | 0,500 | 363,15 | -1790,92 | 0,300 | 293,15 | -4735,89 | 0,469 | -4447,70 | 300,000 | 0,000487 | 1,83019 |
| 13 | 70 | 1,034 | 0,500 | 363,15 | -1790,92 | 0,300 | 293,15 | -4735,89 | 0,477 | -4319,59 | 303,045 | 0,000475 | 1,62378 |
| 14 | 140 | 1,067 | 0,500 | 363,15 | -1790,92 | 0,300 | 293,15 | -4735,89 | 0,485 | -4205,92 | 305,747 | 0,000464 | 1,44760 |
| 15 | 210 | 1,100 | 0,500 | 363,15 | -1790,92 | 0,300 | 293,15 | -4735,89 | 0,492 | -4104,59 | 308,155 | 0,000453 | 1,29622 |
| 16 | 280 | 1,132 | 0,500 | 363,15 | -1790,92 | 0,300 | 293,15 | -4735,89 | 0,499 | -4013,86 | 310,312 | 0,000443 | 1,16533 |
| 17 | 350 | 1,163 | 0,500 | 363,15 | -1790,92 | 0,300 | 293,15 | -4735,89 | 0,506 | -3932,28 | 312,251 | 0,000433 | 1,05153 |
| 18 | 420 | 1,193 | 0,500 | 363,15 | -1790,92 | 0,300 | 293,15 | -4735,89 | 0,513 | -3858,68 | 314,001 | 0,000423 | 0,95207 |
| 19 | 490 | 1,223 | 0,500 | 363,15 | -1790,92 | 0,300 | 293,15 | -4735,89 | 0,519 | -3792,03 | 315,585 | 0,000414 | 0,86472 |
| 20 | 560 | 1,252 | 0,500 | 363,15 | -1790,92 | 0,300 | 293,15 | -4735,89 | 0,525 | -3731,50 | 317,024 | 0,000405 | 0,78768 |
| …. | …. | ….. | ….. | …… | …… | …… | …… | …… | ……. | ……. | …… | …… | ….. |
| 300 | 20160 | 2,879 | 0,500 | 363,15 | -1790,92 | 0,300 | 293,15 | -4735,89 | 0,796 | -2895,33 | 336,899 | 0,000005 | 0,00002 |


Temperaturas de entradas y salida
400
390
380
370
360
350
340
330
320
310
300
290
280
270
260
250
Temperaturas [ºK]
T1
T2
Ts
0
2000
4000
6000
8000
10000
12000
14000
16000
18000
20000
Tiempo [s]


Flujos de entradas y salidas
0.900
0.800
0.700
0.600
0.500
0.400
0.300
0.200
0.100
0.000
Flujos [Kgmol/s]
F1
F2
Fs
0
2000
4000
6000
8000
10000
12000
14000
16000
18000
20000
Tiempo [s]




## b- Dos componentes:

$$\boxed{\frac{dh}{dt} = \frac{(F1 + F2 - Fs)}{\rho_S \times AT}}$$

$$Ao = \frac{\pi \times Do^2}{4}$$
$$AT = \frac{\pi \times DT^2}{4}$$

$$\boxed{Fs = \rho_S \times Ao \times \sqrt{2 \times g \times h}}$$

$$\boxed{\begin{array}{l}\dfrac{dx_1}{dt}=\dfrac{\left[\dfrac{F1-x_1\times Fs}{AT\times\rho_S}-x_1\times\dfrac{dh}{dt}\right]}{h}\\ \dfrac{dx_2}{dt}=\dfrac{\left[\dfrac{F2-x_2\times Fs}{AT\times\rho_S}-x_2\times\dfrac{dh}{dt}\right]}{h}\end{array}}$$

$$\frac{d(H_S\times M)}{dt}=F1\times H_1+F2\times H_2-Fs\times H_S$$

$$\frac{d(H_S\times h\times AT\times\rho_S)}{dt}=F1\times H_1+F2\times H_2-Fs\times H_S$$

$$AT\times\rho_S\times\frac{d(H_S\times h)}{dt}=F1\times H_1+F2\times H_2-Fs\times H_S$$

$$\frac{d(H_S\times h)}{dt}=\frac{F1\times H_1+F2\times H_2-Fs\times H_S}{AT\times\rho_S}$$

$$h\frac{d(H_S)}{dt}+H_S\frac{d(h)}{dt}=\frac{F1\times H_1+F2\times H_2-Fs\times H_S}{AT\times\rho_S}$$

$$\boxed{\frac{d(H_S)}{dt}=\left[\frac{F1\times H_1+F2\times H_2-Fs\times H_S}{AT\times\rho_S}-H_S\frac{d(h)}{dt}\right]\Big/h}$$

$$\boxed{\begin{array}{c}H_1=a_1\times T_1+b_1\\ H_2=a_2\times T_2+b_2\\ H_S=\dfrac{F1\times(a_1\times T_S+b_1)+F2\times(a_2\times T_S+b_2)}{F_S}\end{array}}$$

$$\boxed{\rho_S=x_1\times\rho_1+x_2\times\rho_2}$$

Caso de aplicación




Parámetros del problema:

F1: 0,06 [kmol/s]
T1= 363,15 [ºK]
ρ1: 11,186 [mol/m$^3$]
a1= 133,296 [KJ/Kg mol ºK]
b1= 9691,2069 [KJ/Kg mol]
F2: 0,10 [kmol/s]
P2: 9,363 [mol/m$^3$]
T2= 293,15 [ºK]
a2= 160,5239 [KJ/Kg mol ºK]
b2= -35832,1603 [KJ/Kg mol]

DT: 4 [m]
Do: 0.05 [m]
HT: 5 [m]
AT: 12,566 [m$^2$]
Ao: 1,9635E-03 [m$^2$]

Condiciones iniciales:

Ts = 300 ºK → HS= 2,880E+04 KJ/Kgmol
h = 1 [m]
x(1)=1,0000
x(2)=0,0000

### *Resolución algebraica a tiempo infinito (estado estacionario):*

$$Fs = F1 + F2 = 0{,}06\ [Kgmol/s] + 0{,}1\ [Kgmol/s] = 0{,}16\ [Kgmol/s]$$

$$x_1 = \frac{F1}{F1+F2} = 0{,}375$$

$$x_2 = \frac{F2}{F1+F2} = 0{,}625$$

$$H_1 = a_1 \times 363{,}15 + b_1 = 5{,}810 \times 10^4\ [KJ/kgmol]$$

$$H_2 = a_2 \times 293{,}15 + b_2 = 1{,}123 \times 10^4 [KJ/kgmol]$$

$$H_S = \frac{F1 \times H_1 + F2 \times H_2}{F_S} = 2{,}880 \times 10^4 \left[ KJ / kgmol \right]$$

$$H_S = (x_1 \times a_1 + x_2 \times a_2) \times T_S + (b_1 + b_2)$$

$$T_S = \frac{H_S - (b_1 + b_2)}{x_1 \times a_1 + x_2 \times a_2} = 316{,}428 \;\; \left[ {}^{\circ}\mathrm{K} \right]$$

$$\rho_S = 0.375 \times 11.1856 \left[ \frac{\mathrm{kmol}}{\mathrm{m}^3} \right] + 0.625 \times 9.3632 \left[ \frac{\mathrm{kmol}}{\mathrm{m}^3} \right] = 10{,}047 \left[ \frac{\mathrm{kmol}}{\mathrm{m}^3} \right]$$

$$h_{EE} = \frac{\left( \dfrac{F1 + F2}{\rho_S \times Ao} \right)^2}{2 \times g} = \frac{\left( \dfrac{0{,}06 + 0{,}1}{10{,}047 \times 1{,}9635 \times 10^{-2}} \right)^2}{2 \times 9{,}8} = 3{,}348 \left[ m \right]$$

### *Resolución con simulador EMSO*

```
#* Balance *#

using "types";

FlowSheet Balance

PARAMETERS
F1 as Real (Brief="flujo molar entrada 1",Unit='kmol/s');
F2 as Real (Brief="flujo molar entrada 2",Unit='kmol/s');
DT as Real (Brief="Diametro tanque", Unit='m');
g as Real (Brief="Ac de la gravedad",Unit='m/s^2',
Default=9.8);
Do as Real (Brief="Diametro orificio",Unit='m');
pi as Real (Brief="Cte pi", Default=3.1415923536);
T1 as Real (Brief="Temperatura entrada 1",Unit='K');
T2 as Real (Brief="Temperatura entrada 2",Unit='K');
a(2) as Real(Brief="param entalpia
molar",Unit='kJ/kmol/K');
b(2) as Real(Brief="param entalpia molar",Unit='kJ/kmol');
Ro1 as Real(Brief="densidad entrada 1", Unit='kmol/m^3');
Ro2 as Real(Brief="densidad entrada 1", Unit='kmol/m^3');

VARIABLES
Fs as Real (Brief="flujo molar salida",Unit='kmol/s');
h as Real(Brief="altura",Unit='m');
H1 as Real (Brief="entalpia entrada 1",Unit='kJ/kmol');
```

```
H2 as Real (Brief="entalpia entrada 2",Unit='kJ/kmol');
Hs as Real (Brief="entalpia salida",Unit='kJ/kmol');
Ts as Real (Brief="Temperatura salida",Unit='K');
AT as Real (Brief="Area tanque", Unit='m^2');
Ao as Real (Brief="Area orificio", Unit='m^2');
x(2) as Real(Brief="comp salida");
RoS as Real(Brief="densidad salida", Unit='kmol/m^3');

EQUATIONS
AT=pi*DT^2/4;
Ao=pi*Do^2/4;
diff(h)=(F1+F2-Fs)/(AT*RoS);
Fs=RoS*Ao*sqrt(2*g*h);
diff(h*Hs)=(F1*H1+H2*F2-Hs*Fs)/(RoS*AT);
H1=a(1)*T1+b(1);
H2=a(2)*T2+b(2);
Hs=(x(1)*a(1)+x(2)*a(2))*Ts+(x(1)*b(1)+x(2)*b(2));
diff(h*x(1))=(F1-Fs*x(1))/(AT*RoS);
diff(h*x(2))=(F2-Fs*x(2))/(AT*RoS);
RoS=x(1)*Ro1+x(2)*Ro2;

SET
F1=0.06*'kmol/s';
T1=363.15*'K';
Ro1=11.18568232*'kmol/m^3';
a(1)=133.2962*'kJ/kmol/K';
b(1)=9691.2069*'kJ/kmol';

F2=0.1*'kmol/s';
T2=293.15*'K';
Ro2=9.363295880*'kmol/m^3';
a(2)=160.5239*'kJ/kmol/K';
b(2)=-35832.1603*'kJ/kmol';

DT=4*'m';
Do=0.05*'m';

INITIAL
h=1*'m';
Hs=49680*'kJ/kmol';
x(1)=1;
x(2)=0.00001;

GUESS

OPTIONS
TimeStart=0;
TimeEnd=20000;
```

```
TimeStep=100;
TimeUnit='s';
DAESolver(File="dassl");
Dynamic=true;

end
```








T1 ['K']
T2 ['K']
Ts ['K']
370
360
350
340
330
320
310
300
290
0.00
2500
5000
7500
1.000x10^4
1.250x10^4
1.500x10^4
1.750x10^4
2.000x10^4




x1
x2
1.00
0.800
0.600
0.400
0.200
0.00
0.00
2500
5000
7500
1.000x10^4
1.250x10^4
1.500x10^4
1.750x10^4
2.000x10^4

### *Resolución con planillas Excel*

Implementar el siguiente modelo para su resolución en estado estacionario:

| | A | B | C | D | E | F |
|---|---|---|---|---|---|---|
| 1 | Corriente | F1 | F2 | | Fs | Unidades |
| 2 | Flujo masico | 0,06 | 0,1 | | =B2+C2 | Kgmol/s |
| 3 | Densidad | 11,18568232 | 9,36329588 | | =B3*E8+C3*E9 | Kgmol/m3 |
| 4 | Temperatura | 363,15 | 293,15 | | =(E7-E6)/E5 | ºK |
| 5 | a | 133,2962 | 160,5239 | | =B5*E8+C5*E9 | KJ/Kgmol ºK |
| 6 | b | 9691,2069 | - 35832,1603 | | =E8*B6+E9*C6 | KJ/Kgmol |
| 7 | Entalpia | =$B$5*B4+$B$6 | =C5*C4+C6 | | =(B2*B7+C2*C7)/E2 | KJ/Kgmol |
| 8 | x1 | 1 | 0 | | =B2*B8/E2 | adim |
| 9 | x2 | =1-B8 | =1-C8 | | =1-E8 | adim |

| | H | I | J |
|---|---|---|---|
| 1 | Param | Valor | Unidades |
| 2 | Ro | =E3 | kmol/m^3 |
| 3 | Do | 0,05 | m |
| 4 | Ao | =PI()*I3*I3/4 | $m^2$ |
| 5 | DT | 4 | m |
| 6 | AT | =PI()*I5*I5/4 | $m^2$ |
| 7 | Hee | =((E2/(I2*I4))^2)/(2*9,8) | m |

Que resuelta queda:

| | A | B | C | D | E | F | G | H | I | J |
|---|---|---|---|---|---|---|---|---|---|---|
| 1 | Corriente | F1 | F2 | | Fs | Unidades | | Param | Valor | Unidades |
| 2 | Flujo masico | 0,06000 | 0,10000 | | 0,16 | Kgmol/s | | Ro | 10,047 | kmol/m^3 |
| 3 | Densidad | 11,186 | 9,363 | | 10,047 | Kgmol/m3 | | Do | 0,05 | m |
| 4 | Temperatura | 363,150 | 293,150 | | 316,428 | ºK | | Ao | 1,9635E-03 | $m^2$ |
| 5 | a | 133,296 | 160,5239 | | 150,314 | KJ/Kgmol ºK | | DT | 4 | m |
| 6 | b | 9691,2069 | -35832,1603 | | -18760,8976 | KJ/Kgmol | | AT | 12,566 | $m^2$ |
| 7 | Entalpia | 5,810E+04 | 1,123E+04 | | 2,880E+04 | KJ/Kgmol | | Hee | 3,3564 | m |
| 8 | x1 | 1,000 | 0,000 | | 0,375 | adim | | | | |
| 9 | x2 | 0,000 | 1,000 | | 0,625 | adim | | | | |

Para resolverlo en estado dinámico se insertan las siguientes celdas:

| | A | B | C | D | E | F | G | H |
|---|---|---|---|---|---|---|---|---|
| 10 | | | | | | | | |
| 11 | dt | 100 | | | | hf | =B302 | Fsf |
| 12 | | | | | | | | |
| 13 | t | h | F1 | T1 | H1 | F2 | T2 | H2 |
| 14 | 0 | 1 | =B2 | =$B$4 | =D14*$B$5+$B$6 | =C2 | =$C$4 | =G14*$C$5+$C$6 |
| 15 | =A14+$B$11 | =B14+$B$11*L14 | =C14 | =D14 | =D15*$B$5+$B$6 | =F14 | =G14 | =G15*$C$5+$C$6 |
| 16 | =A15+$B$11 | =B15+$B$11*L15 | =C15 | =D15 | =D16*$B$5+$B$6 | =F15 | =G15 | =G16*$C$5+$C$6 |
| 17 | =A16+$B$11 | =B16+$B$11*L16 | =C16 | =D16 | =D17*$B$5+$B$6 | =F16 | =G16 | =G17*$C$5+$C$6 |
| 18 | =A17+$B$11 | =B17+$B$11*L17 | =C17 | =D17 | =D18*$B$5+$B$6 | =F17 | =G17 | =G18*$C$5+$C$6 |
| 19 | =A18+$B$11 | =B18+$B$11*L18 | =C18 | =D18 | =D19*$B$5+$B$6 | =F18 | =G18 | =G19*$C$5+$C$6 |
| 20 | =A19+$B$11 | =B19+$B$11*L19 | =C19 | =D19 | =D20*$B$5+$B$6 | =F19 | =G19 | =G20*$C$5+$C$6 |
| ….. | …. | …. | …. | ….. | ….. | ….. | ….. | ….. |
| 302 | =A301+$B$11 | =B301+$B$11*L301 | =C301 | =D301 | =D302*$B$5+$B$6 | =F301 | =G301 | =G302*$C$5+$C$6 |

| | I | J | K | L |
|---|---|---|---|---|
| 10 | | | | |
| 11 | =I302 | TSf | =K302 | |
| 12 | | | | |
| 13 | Fs | Hs | Ts | dh |
| 14 | =$E$3*$I$4*RAIZ(2*9,8*B14) | 49680,0669 | =(J14-T14)/S14 | =(C14+F14-I14)/($E$3*$I$6) |
| 15 | =$E$3*$I$4*RAIZ(2*9,8*B15) | =J14+$B$11*M14 | =(J15-T15)/S15 | =(C15+F15-I15)/($E$3*$I$6) |
| 16 | =$E$3*$I$4*RAIZ(2*9,8*B16) | =J15+$B$11*M15 | =(J16-T16)/S16 | =(C16+F16-I16)/($E$3*$I$6) |
| 17 | =$E$3*$I$4*RAIZ(2*9,8*B17) | =J16+$B$11*M16 | =(J17-T17)/S17 | =(C17+F17-I17)/($E$3*$I$6) |
| 18 | =$E$3*$I$4*RAIZ(2*9,8*B18) | =J17+$B$11*M17 | =(J18-T18)/S18 | =(C18+F18-I18)/($E$3*$I$6) |
| 19 | =$E$3*$I$4*RAIZ(2*9,8*B19) | =J18+$B$11*M18 | =(J19-T19)/S19 | =(C19+F19-I19)/($E$3*$I$6) |
| 20 | =$E$3*$I$4*RAIZ(2*9,8*B20) | =J19+$B$11*M19 | =(J20-T20)/S20 | =(C20+F20-I20)/($E$3*$I$6) |
| …. | … | …. | …. | …. |
| 302 | =$E$3*$I$4*RAIZ(2*9,8*B302) | =J301+$B$11*M301 | =(J302-T302)/S302 | =(C302+F302-I302)/($E$3*$I$6) |

| | M | N | O |
|---|---|---|---|
| 10 | | =N302 | =O302 |
| 11 | | | |
| 12 | | | |
| 13 | dHs | x1 | x2 |
| 14 | =((C14*E14+F14*H14-I14*J14)/($I$6*$E$3)-J14*L14)/B14 | =1 | 0 |
| 15 | =((C15*E15+F15*H15-I15*J15)/($I$6*$E$3)-J15*L15)/B15 | =N14+Q14*$B$11 | =O14+R14*$B$11 |
| 16 | =((C16*E16+F16*H16-I16*J16)/($I$6*$E$3)-J16*L16)/B16 | =N15+Q15*$B$11 | =O15+R15*$B$11 |
| 17 | =((C17*E17+F17*H17-I17*J17)/($I$6*$E$3)-J17*L17)/B17 | =N16+Q16*$B$11 | =O16+R16*$B$11 |
| 18 | =((C18*E18+F18*H18-I18*J18)/($I$6*$E$3)-J18*L18)/B18 | =N17+Q17*$B$11 | =O17+R17*$B$11 |
| 19 | =((C19*E19+F19*H19-I19*J19)/($I$6*$E$3)-J19*L19)/B19 | =N18+Q18*$B$11 | =O18+R18*$B$11 |
| 20 | =((C20*E20+F20*H20-I20*J20)/($I$6*$E$3)-J20*L20)/B20 | =N19+Q19*$B$11 | =O19+R19*$B$11 |
| .... | .... | .... | .... |
| 302 | =((C302*E302+F302*H302-I302*J302)/($I$6*$E$3)-J302*L302)/B302 | =N301+Q301*$B$11 | =O301+R301*$B$11 |

| | P | Q |
|---|---|---|
| 10 | | |
| 11 | | |
| 12 | | |
| 13 | RoS | dx1 |
| 14 | =N14*$B$3+O14*$C$3 | =((C14-N14*I14)/($I$6*P14)-N14*L14)/B14 |
| 15 | =N15*$B$3+O15*$C$3 | =((C15-N15*I15)/($I$6*P15)-N15*L15)/B15 |
| 16 | =N16*$B$3+O16*$C$3 | =((C16-N16*I16)/($I$6*P16)-N16*L16)/B16 |
| 17 | =N17*$B$3+O17*$C$3 | =((C17-N17*I17)/($I$6*P17)-N17*L17)/B17 |
| 18 | =N18*$B$3+O18*$C$3 | =((C18-N18*I18)/($I$6*P18)-N18*L18)/B18 |
| 19 | =N19*$B$3+O19*$C$3 | =((C19-N19*I19)/($I$6*P19)-N19*L19)/B19 |
| 20 | =N20*$B$3+O20*$C$3 | =((C20-N20*I20)/($I$6*P20)-N20*L20)/B20 |
| .... | .... | .... |
| 302 | =N302*$B$3+O302*$C$3 | =((C302-N302*I302)/($I$6*P302)-N302*L302)/B302 |

| | R | S | T |
|---|---|---|---|
| 10 | | | |
| 11 | | | |
| 12 | | | |
| 13 | dx2 | aS | bS |
| 14 | =((F14-O14*I14)/($I$6*P14)-(O14*L14))/B14 | =N14*$B$5+O14*$C$5 | =N14*$B$6+O14*$C$6 |
| 15 | =((F15-O15*I15)/($I$6*P15)-(O15*L15))/B15 | =N15*$B$5+O15*$C$5 | =N15*$B$6+O15*$C$6 |
| 16 | =((F16-O16*I16)/($I$6*P16)-(O16*L16))/B16 | =N16*$B$5+O16*$C$5 | =N16*$B$6+O16*$C$6 |
| 17 | =((F17-O17*I17)/($I$6*P17)-(O17*L17))/B17 | =N17*$B$5+O17*$C$5 | =N17*$B$6+O17*$C$6 |
| 18 | =((F18-O18*I18)/($I$6*P18)-(O18*L18))/B18 | =N18*$B$5+O18*$C$5 | =N18*$B$6+O18*$C$6 |
| 19 | =((F19-O19*I19)/($I$6*P19)-(O19*L19))/B19 | =N19*$B$5+O19*$C$5 | =N19*$B$6+O19*$C$6 |
| 20 | =((F20-O20*I20)/($I$6*P20)-(O20*L20))/B20 | =N20*$B$5+O20*$C$5 | =N20*$B$6+O20*$C$6 |
| .... | ... | ... | ... |
| 302 | =((F302-O302*I302)/($I$6*P302)-(O302*L302))/B302 | =N302*$B$5+O302*$C$5 | =N302*$B$6+O302*$C$6 |

Que resuelto queda:

| | A | B | C | D | E | F | G | H | I | J | K |
|---|---|---|---|---|---|---|---|---|---|---|---|
| 10 | | | | | | | | | | | |
| 11 | | | | hf | 3,348 | Fsf | 0,160 | TSf | 316,43 | | |
| 12 | | | | | | | | | | | |
| 13 | F1 | T1 | H1 | F2 | T2 | H2 | Fs | Hs | Ts | dh | dHs |
| 14 | 0,060 | 363,15 | 58097,72 | 0,100 | 293,15 | 11225,42 | 0,087 | 49680,07 | 300,000 | 0,000576 | -26,45856 |
| 15 | 0,060 | 363,15 | 58097,72 | 0,100 | 293,15 | 11225,42 | 0,090 | 47034,21 | 302,252 | 0,000556 | -21,84790 |
| 16 | 0,060 | 363,15 | 58097,72 | 0,100 | 293,15 | 11225,42 | 0,092 | 44849,42 | 304,156 | 0,000537 | -18,26936 |
| 17 | 0,060 | 363,15 | 58097,72 | 0,100 | 293,15 | 11225,42 | 0,094 | 43022,48 | 305,762 | 0,000520 | -15,44370 |
| 18 | 0,060 | 363,15 | 58097,72 | 0,100 | 293,15 | 11225,42 | 0,096 | 41478,11 | 307,117 | 0,000504 | -13,17905 |
| 19 | 0,060 | 363,15 | 58097,72 | 0,100 | 293,15 | 11225,42 | 0,098 | 40160,21 | 308,264 | 0,000488 | -11,34026 |
| 20 | 0,060 | 363,15 | 58097,72 | 0,100 | 293,15 | 11225,42 | 0,100 | 39026,18 | 309,238 | 0,000473 | -9,83005 |
| .... | ... | ... | .... | ... | ... | ... | ... | ... | ... | ... | ... |
| 302 | 0,060 | 363,15 | 58097,72 | 0,100 | 293,15 | 11225,42 | 0,104 | 37185,48 | 310,783 | 0,000446 | -7,52782 |

| | L | M | N | O | P | Q | R | S | T |
|---|---|---|---|---|---|---|---|---|---|
| 10 | | | | | | | | | |
| 11 | | | 0,375 | 0,625 | | | | | |
| 12 | | | | | | | | | |
| 13 | dh | dHs | x1 | x2 | RoS | dx1 | dx2 | aS | bS |
| 14 | 0,000576 | -26,45856 | 1 | 0 | 11,1856823 | -0,000770031 | 0,000711423 | 133,2962 | 9691,206 |
| 15 | 0,000556 | -21,84790 | 0,922997 | 0,071142 | 10,9904762 | -0,000641965 | 0,000603508 | 134,4520135 | 6395,77316 |
| 16 | 0,000537 | -18,26936 | 0,8588 | 0,131493 | 10,8374763 | -0,000540873 | 0,000516227 | 135,5826012 | 3611,13203 |
| 17 | 0,000520 | -15,44370 | 0,804713 | 0,183116 | 10,7158317 | -0,000459947 | 0,000444848 | 136,6596475 | 1237,20837 |
| 18 | 0,000504 | -13,17905 | 0,758718 | 0,227601 | 10,6178742 | -0,000394361 | 0,000385898 | 137,6696071 | -802,523136 |
| 19 | 0,000488 | -11,34026 | 0,719282 | 0,26619 | 10,5380828 | -0,00034062 | 0,000336782 | 138,6075191 | -2567,4632 |
| 20 | 0,000473 | -9,83005 | 0,68522 | 0,299869 | 10,4724146 | -0,000296151 | 0,00029553 | 139,4733331 | -4104,3268 |
| …. | … | … | … | … | … | … | … | … | … |
| 302 | 0,000002 | -0,00001 | 0,375001 | 0,624999 | 10,0466913 | -3,81505E-10 | 4,34466E-10 | 150,3134626 | -18760,846 |


Nivel de líquido en el tanque
Nivel liquido [m]
4.00
3.50
3.00
2.50
2.00
1.50
1.00
0.50
0.00
0
5000
10000
15000
20000
25000
Tiempo [s]


Temperaturas de entradas y salida
Temperaturas [ºK]
400
390
380
370
360
350
340
330
320
310
300
290
280
270
260
250
0
5000
10000
15000
20000
25000
Tiempo [s]
T1
T2
Ts



Composición de salida
Fracción molar
Tiempo [s]
x1
x2
1.000
0.900
0.800
0.700
0.600
0.500
0.400
0.300
0.200
0.100
0.000
0
5000
10000
15000
20000
25000